



## Solution

SP4R03s03.sas

Use the **Cars** data set in the **SP4R** library to complete this exercise.

### 1. Creating and Using a SAS Function

- a. Create a function called **tier** with a single numeric argument, which returns a character value. The function should return values according to the following table:

input	output
<20	Low
20-29	Medium
>30	High

```
proc fcmp outlib=work.functions.newfuncs;
  function tier(val) $;
    length newval $ 6;
    if val < 20 then newval = 'Low';
    else if val <30 then newval='Medium';
    else newval='High';
    return(newval);
  endsub;
quit;
```

- b. Use the function that you created to create a new variable in the **Cars** data set. Name the new variable **mpg\_quality2** and name the argument of the function **tier** as **mpg\_average**. As a result, **mpg\_quality** and **mpg\_quality2** are identical.

```
options cmplib=work.functions;
data sp4r.cars;
  set sp4r.cars;
  mpg_quality2=tier(mpg_average);
run;
```

- c. Print observations 65 through 70 for the variables **mpg\_average**, **mpg\_quality**, and **mpg\_quality2** to ensure that the variable is created.

```
proc print data=sp4r.cars (firstobs=65 obs=70);
  var mpg_average mpg_quality mpg_quality2;
run;
```

	Obs	mpg_ average	mpg_ quality	mpg_ quality2
	65	16.0	Low	Low
	66	18.5	Low	Low
	67	20.5	Medium	Medium
	68	31.0	High	High
	69	31.0	High	High